```
In[•]:= (* ::Package::*)(*Title:Einstein Tensor Calculation for 3+1 Conformally Flat Metric*)
(*10mm Wormhole*)Print["Defining the potential and metric components..."];

(*Parameters for 10mm wormhole*)
R0 = 2.0
w = 0.2
A = 0.01
ε = 0.04

(*Potential function Φ(x,y,z)*)
rCoord[x_, y_, z_] := Max[Sqrt[x^2 + y^2 + z^2], ε];
Φ[x_, y_, z_] := -A * (1 - R0 / rCoord[x, y, z]) * Exp[-((rCoord[x, y, z] - R0)^2 / w^2)];

(*Metric components in (t,x,y,z) coordinates*)
(*ds²=-e^{2Φ}dt²+e^{-2Φ}(dx²+dy²+dz²)*)
gtt[x_, y_, z_] := -Exp[2 * Φ[x, y, z]]
gxx[x_, y_, z_] := Exp[-2 * Φ[x, y, z]]
gyy[x_, y_, z_] := Exp[-2 * Φ[x, y, z]]
gzz[x_, y_, z_] := Exp[-2 * Φ[x, y, z]]

(*Metric tensor:g_{μν}*)
metric = {{gtt[x, y, z], 0, 0, 0},
   {0, gxx[x, y, z], 0, 0}, {0, 0, gyy[x, y, z], 0}, {0, 0, 0, gzz[x, y, z]}};

Print["Metric tensor at throat (x=R0,y=0,z=0):"];
MatrixForm[metric /. {x → R0, y → 0, z → 0}]

(*Coordinate system*)
Coordinates = {t, x, y, z};

(*Inverse metric*)
inversemetric = Inverse[metric];

(*Christoffel symbols Γ^μ_{νρ}*)
Print["\nCalculating Christoffel symbols..."];
Clear[Γ]
Γ[μ_, ν_, ρ_] := Sum[(1 / 2) * inversemetric⟦μ, σ⟧ * (D[metric⟦σ, ν⟧, Coordinates⟦ρ⟧] +
      D[metric⟦σ, ρ⟧, Coordinates⟦ν⟧] - D[metric⟦ν, ρ⟧, Coordinates⟦σ⟧]), {σ, 1, 4}];

(*Riemann tensor:R^μ_{νρσ}*)
Print["\nCalculating Riemann tensor..."];
Clear[Riemann]
Riemann[μ_, ν_, ρ_, σ_] := D[Γ[μ, ν, σ], Coordinates⟦ρ⟧] - D[Γ[μ, ν, ρ], Coordinates⟦σ⟧] +
   Sum[Γ[μ, κ, ρ] * Γ[κ, ν, σ] - Γ[μ, κ, σ] * Γ[κ, ν, ρ], {κ, 1, 4}];

(*Ricci tensor:R_{μν}*)
```

```
Print["\nCalculating Ricci tensor and scalar..."];
Clear[Ricci]
Ricci[μ_, ν_] := Sum[Riemann[σ, μ, σ, ν], {σ, 1, 4}];

RicciScalar = Sum[inversemetric⟦μ, ν⟧ * Ricci[μ, ν], {μ, 1, 4}, {ν, 1, 4}];

(*Einstein tensor:G_{μν}*)
Print["\nCalculating Einstein tensor..."];
Clear[Einstein]
Einstein[μ_, ν_] := Ricci[μ, ν] - (1 / 2) * metric⟦μ, ν⟧ * RicciScalar;

(*Compute at throat for analysis*)
x0 = R0; y0 = 0; z0 = 0;
Print["\nEinstein tensor at throat x=", x0, ", y=", y0, ", z=", z0, ":"];

einsteinAtThroat = Table[Einstein[i, j] /. {x → x0, y → y0, z → z0}, {i, 1, 4}, {j, 1, 4}];
MatrixForm[Simplify[einsteinAtThroat]]

(*ADM mass estimate*)
Print["\nADM Mass estimation..."];
PhiRadial[r_] := -A * (1 - R0 / r) * Exp[-((r - R0)^2 / w^2)]
ADMmass = Limit[r^2 * D[PhiRadial[r], r], r → Infinity]
Print["ADM Mass: ", ADMmass]

Print["\nCalculation complete."]
```

```
Defining the potential and metric components...
```

Out[•]=

```
2.
```

Out[•]=

```
0.2
```

Out[•]=

```
0.01
```

Out[•]=

```
0.04
```

```
Metric tensor at throat (x=R0,y=0,z=0):
```

Out[•]//MatrixForm=

$$\begin{pmatrix} -1. & 0 & 0 & 0 \\ 0 & 1. & 0 & 0 \\ 0 & 0 & 1. & 0 \\ 0 & 0 & 0 & 1. \end{pmatrix}$$

Calculating Christoffel symbols...

Calculating Riemann tensor...

Calculating Ricci tensor and scalar...

Calculating Einstein tensor...

Einstein tensor at throat x=2., y=0, z=0:

Out[•]//MatrixForm=

$$\begin{pmatrix} -0.000025 & 0. & 0. & 0. \\ 0. & -0.000025 & 0. & 0. \\ 0. & 0. & 0.000025 & 0. \\ 0. & 0. & 0. & 0.000025 \end{pmatrix}$$

ADM Mass estimation...

Out[•]=

0.

ADM Mass: 0.

Calculation complete.